

Загрузчик для NeoGS v 0.2

Загрузчик предназначен для загрузки прошивки (вместо GS105a.ROM зашитой в ПЗУ) со стороны Спекы, со встроенной в NeoGS SD карты, загрузки кода в любую страницу ОЗУ со Спекы или SD карточки и запуск на выполнение с произвольного адреса в произвольной странице, получение версии фпга в полностью текстовом виде, проверка CRC16 прошиваемого блока и его заголовка.

Последовательность работы загрузчика

- после программного сброса со стороны Спекы делается задержка с ожиданием определенных байт через порты команд и данных.

- при превышении лимита времени ожидания делается попытка загрузить прошивку (вместо зашитой в ПЗУ) со встроенной SD карты. Файл для загрузки должен находиться на SD карте и располагаться по следующему пути:

SYSTEMZX\NEOGS\NEOGS.ROM

Все имена должны быть короткими, длинные имена не поддерживаются.

- если файл не найден, то загружается и запускается прошивка зашитая в ПЗУ (на данный момент это версия GS105a.ROM)

Последовательность доступа к функциям загрузчика

```
LD A,#80
OUT (GSCTR),A ;сброс NeoGS
NOP
NOP
NOP ;ждем 12 тактов
LD A,#55
OUT (GSDAT),A
OUT (GSCOM),A ;первый ожидаемый байт загрузчиком
CALL WC ;ждем очистки COMAND BIT
LD A,#AA
OUT (GSDAT),A
OUT (GSCOM),A ;второй ожидаемый байт загрузчиком
CALL WC
;здесь сброса COMAND BIT ждать несколько дольше пока переносится
;код загрузчика из ПЗУ в ОЗУ
;частота работы процессора после переноса кода устанавливается 12МГц
;при загрузке с SD карты временно устанавливается частота 24МГц
```

Команды загрузчика доступные в этой версии

Команды и параметры загрузчику подаются также, как и в стандартной прошивке GS105a. Сначала записывается байт параметра в порт данных (GSDAT), потом записывается байт в порт команд (GSCOM). Далее остальные байты записываются в порт данных, если они есть. Вызываемые подпрограммы WC, WD, WN такие же как и для стандартной прошивки.

```
WC      IN A, (GSCOM)
        RRCA
        JR C,WC
        RET
```

```

WD      IN A, (GSCOM)
        RLCA
        JR C,WD
        RET

```

```

WN      IN A, (GSCOM)
        RLCA
        JR NC,WN
        RET

```

#00 — загрузить со Спека 32 Кб прошивки вместо загрузки из ПЗУ. Загружает в страницу 0 в адрес 0 ОЗУ. Команда только загружает прошивку, но не запускает.

```

        XOR A
        OUT (GSCOM), A
        CALL WC
        LD HL,адрес откуда передается прошивка
        LD DE,#8000      ;длина передачи
        LD BC,GSDAT
LOOP    OUTI
        CALL WD
        DEC DE
        LD A,D
        OR E
        JR NZ,LOOP

```

#01 — запустить прошивку. Устанавливает защиту от записи в странице 0 ОЗУ, куда загружена прошивка. Также выставляет частоту процессора 12МГц. Без параметров.

#02 — загружает и запускает прошивку из ПЗУ. Полностью соответствует старому загрузчику. Без параметров.

#03 — загружает и запускает прошивку с SD карты. Если найдет. Без параметров.

#04 — загрузка кода со Спека. Загружает в любую страницу ОЗУ в адрес #8000 (32768) до 32Кб. Для загрузки в другую страницу функцию надо вызывать еще раз.

```

        LD A,номер страницы
        OUT (GSDAT), A
        LD A,4
        OUT (GSCOM), A
        CALL WC
        LD BC,GSDAT
        LD DE,размер блока (не более 32768 байт)
        LD HL,адрес откуда слать
        OUT (C),E
        CALL WD
        OUT (C),D
        CALL WD
LOOP    OUTI
        CALL WD
        DEC DE
        LD A,D
        OR E
        JR NZ,LOOP

```

#05 — загрузка произвольного файла с SD карты по указанному пути. Загружает в любую страницу

ОЗУ по адресу #8000 (32768), кроме страниц 0 и 1. Для страниц 02...#7F загрузка будет производиться пока не закончатся страницы памяти или до окончания файла.

Ограничения на загрузку. В страницу 0 можно загрузить не более 32Кб. Загрузка в страницу 1 запрещена, статус загружаемого файла будет #AA-не найден. Статус загруженного файла проверять функцией 7.

```
LD A,номер страницы
OUT (GSDAT),A
LD A,5
OUT (GSCOM),A
CALL WC
LD HL,адрес текстовой строки
LD BC,GSDAT
LOOP LD A,(HL)
OUTI
EX AF,AF'
CALL WD
EX AF,AF'
AND A
JR NZ,LOOP
```

#06 – запуск кода в любой странице по любому адресу.

```
LD A,номер страницы
OUT (GSDAT),A
LD A,6
OUT (GSCOM),A
CALL WC
LD HL,адрес запуска
LD A,L
OUT (GSDAT),A
CALL WD
LD A,H
OUT (GSDAT),A
CALL WD
```

#07 – получить статус последнего загруженного файла. Без параметров.

Возвращает следующие байты ответа:

#00-файл загружен.

#AA-файл не найден.

#DD-FAT не обнаружен.

#EE-SD не прошла инициализацию или отсутствует.

```
LD A,7
OUT (GSCOM),A
CALL WC
IN A,(GSDAT)
```

#08 – получить распакованную текстовую строку версии страницы.

```
LD A,номер страницы
OUT (GSDAT),A
LD A,8
OUT (GSCOM),A
CALL WC
LD HL,куда положить текст
```

```

LD BC,GSDAT
LOOP CALL WN
IN A,(C)
LD (HL),A
INC HL
AND A
JR NZ,LOOP

```

#09 – получить CRC16 заголовка, блока или количество загруженных блоков для прошивания. Подать в порт данных байт, получить 3 байта ответа.

НЕ РЕКОМЕНДУЕТСЯ ДЛЯ ПРИМЕНЕНИЯ, ИСПОЛЬЗУЕТСЯ ПРИ ПРОШИВАНИИ НОВЫХ ПРОШИВОК.

По записи в порт данных:

бит 7=1 — получить количество блоков для прошивания;

бит 7=0 — число от 0 до 7 выбранный блок в заголовке, остальные биты игнорируются

По чтению ответа:

1 – тип ошибки:

#80-все ОК, следующие 2 байта CRC16 запрошенного блока или количество блоков для прошивания. Иначе последние 2 байта случайные.

#81-CRC16 заголовка ERROR

#82-CRC16 запрошенного блока ERROR

2 – младший байт CRC16 или количества блоков

3 – старший байт CRC16 или количества блоков

```

LD A,байт
OUT (GSDAT),A
LD A,9
OUT (GSCOM),A
CALL WC
IN A,(GSDAT) ;код ошибки
EX AF,AF'
CALL WN
IN A,(GSDAT) ;младший байт ответа
LD L,A
CALL WN
IN A,(GSDAT) ;старший байт ответа
LD H,A
EX AF,AF'

```

#1D - находится несколько за пределами команд загрузчика, используется по аналогии с командой #1D прошивки GS105a.ROM. Предназначена для проверки загрузчик ли сейчас на NeoGS. По этой команде возвращает через порт данных число #76. Для примера: прошивка GS105a.ROM возвращает значение #8B, мой MP3 плеер возвращает #77, когда загружен.

```

XOR A
OUT (GSDAT),A
LD A,#1D
OUT (GSCOM),A
CALL WC
IN A,(GSDAT)

```

Все остальные команды, не описанные выше, перезапускают NeoGS со стандартной прошивкой GS105a защитой в ПЗУ.